
ABSTRACT 
 
Renaming SAS® variables may sound easy, but what if you have hundreds of variables 
that must be renamed? In this instance it will become an annoying and error-prone 
process if you simply use Rename statements to rename each variable individually. This 
paper addresses how you can perform the renaming of many variables quickly and error-
free.  
 
Our Renaming Tool Series is a group of “renaming” macros, which perform different 
renaming jobs, i.e. renaming all the variables in a SAS data set, renaming partial 
variables, adding prefixes, adding suffixes, and/or replacing the prefix or the suffix. It’s a 
very handy tool for tailoring variable names in batches quickly to fit different needs. 
 
INTRODUCTION 
 
Renaming SAS® variables happens a lot in our real life, especially in the Data Analysis 
area. Sometimes when we are handling the real data, for either security issues or per 
clients’ requests, variables are needed to be renamed from common names such as “var 
1”, “var 2”, …, “var n”, to new names with prefix or suffix  of subject or types. But with 
hundreds of variables, it becomes very annoying and time consuming to rename them 
individually. Different macros of renaming variables will be discussed here, which add a 
prefix, or replace prefixes and suffixes. SAS DICTIONARY tables are read-only SAS 
data views that contain lists of things that are related to the current SAS session. We can 
retrieve a number of variables and their names from the DICTIONARY tables. PROC 
SQL’s DICTIONARY.TABLES and COLUMNS are utilized to rename all the variables 
in a SAS data set. The same technique can also be used to rename only selected variables. 
The macro function %SYSFUNC allows access by the macro processor to most data step 
functions and several SCL functions, which allows you to access dataset observations. 
The data set functions, OPEN, CLOSE and VARNAME of %SYSFUNC will be used to 
replace the prefix and suffix. 
 
/* Creating a dataset */ 
data A; 
input id $4. pre_var1_post pre_var2_post pre_var3_post; 
datalines; 
i001 1 2 3 
i002 3 4 5 
i003 6 7 8 
i004 9 10 12 
; 
run; 

Tool 1: Add Prefix on All Variables 
    
Extract number of variables from PROC SQL's DICTIONARY.TABLES and the names 
of the variables from DICTIONARY.COLUMNS, and then attach a prefix to each 
variable name. 
 
 
/* Adding Prefix on all variables */ 
 
%macro rename(lib,dsn,newname); 
proc contents data=&lib..&dsn; 
title 'before renaming'; 
run; 
 
proc sql noprint; 
 select nvar into :num_vars 
 from dictionary.tables 
 where libname="&LIB" and memname="&DSN"; 
 
 select distinct(name) into :var1-
:var%trim(%left(&num_vars))  
 from dictionary.columns 
 where libname="&LIB" and memname="&DSN"; 
 quit; 
run; 
 
proc datasets library = &LIB; 
modify &DSN; 
rename 
%do i = 1 %to &num_vars.; 
&&var&i = &newname._&&var&i. 
%end; 
; 
quit; 
run; 
 
proc contents data=&lib..&dsn.; 
title 'after renaming'; 
run; 
%mend rename; 
 
data B; 
set A; 
run; 
 
%rename(WORK,B,Try1); 

Partial OUTPUT of Tool 1: 
 
   
                                     before renaming  

 
                           Alphabetic List of Variables and Attributes 
 
                                  #    Variable         Type      Len 
 
                                  1    id               Char      4 
                                  2    pre_var1_post    Num       8 
                                  3    pre_var2_post    Num       8 
                                  4    pre_var3_post    Num       8 
 
                                      after renaming 
 
                           Alphabetic List of Variables and Attributes 
 
                                #    Variable              Type     Len 
 
                                1    Try1_id               Char      4 
                                2    Try1_pre_var1_post    Num       8 
                                3    Try1_pre_var2_post    Num       8 
                                4    Try1_pre_var3_post    Num       8 
 
 
 
Tool 2: Add Prefix on Selected Variables 
    
Same approach of renaming all variables but using start and end positions to rename only 
the selected variables. Variable list need to be sorted before running this macro. 
 
/* Adding Prefix on Selected Variables */ 
 
%macro addprefix(lib,dsn,start,end,newname); 
proc contents data=&lib..&dsn; 
title 'before renaming'; 
run; 
 
proc sql noprint; 
 select nvar into :num_vars 
 from dictionary.tables 
 where libname="&LIB" and memname="&DSN"; 
 
 select distinct(name) into :var1-
:var%trim(%left(&num_vars))  
 from dictionary.columns 
 where libname="&LIB" and memname="&DSN"; 
quit; 
run; 
 
proc datasets library = &LIB; 
modify &DSN; 
rename 
%do i = &start. %to &end.; 

&&var&i = New_&&var&i. 
%end; 
; 
quit; 
run; 
 
proc contents data=&lib..&dsn; 
title 'Adding Prefix on Selected variables'; 
run; 
%mend addprefix; 
data c; 
set a; 
run; 
%addprefix(WORK,C,2,4,Try2); 
 
 
 
Partial OUTPUT of Tool 2: 
 
                              Adding Prefix on Selected variables  

 
                           Alphabetic List of Variables and Attributes 
 
                                #    Variable              Type     Len 
 
                                1    id                    Char      4 
                                2    Try2_pre_var1_post    Num       8 
                                3    Try2_pre_var2_post    Num       8 
                                4    Try2_pre_var3_post    Num       8 
 
 
 
Tool 3: Replace Prefix on Selected Variables 
    
The %SYSFUNC macro has allowed access to the SAS component language inside of 
traditional DATA step programming. 
We can easily retrieve variable information by using the data set functions, OPEN, 
CLOSE and VARNAME and replace the existing prefix with the new name.  
 
/* Replacing Prefix on Selected Variables */ 
 
%macro replaceprefix(lib,dsn,start,end,oldprefix,newprefix); 
proc contents data=&lib..&dsn.; 
title 'before renaming'; 
run; 
 
data temp; 
set &lib..&dsn.; 
run; 
 
%LET ds=%SYSFUNC(OPEN(temp,i)); 
%let ol=%length(&oldprefix.); 
%do i=&start %to &end; 

 %let dsvn&i=%SYSFUNC(VARNAME(&ds,&i)); 
 %let l=%length(&&dsvn&i); 
 %let vn&i=&newprefix.%SUBSTR(&&dsvn&i,&ol+1,%EVAL(&l-&ol)); 
%end; 
data &lib..&dsn.; 
set temp; 
  
%do i=&start %to &end; 
&&vn&i=&&dsvn&i; 
drop &&dsvn&i; 
%end; 
%let rc=%SYSFUNC(CLOSE(&ds)); 
proc contents data=&lib..&dsn.; 
title 'Replacing Prefix on Selected variables '; 
run; 
%mend replaceprefix; 
 
data d; 
set a; 
run; 
%replaceprefix(WORK,D,2,4,pre,Try3); 
 
 
 
Partial OUTPUT of Tool 3: 
 
                              Replacing Prefix on Selected variables  

 
                           Alphabetic List of Variables and Attributes 
 
                                  #    Variable          Type     Len 
 
                                  1    id                Char      4 
                                  2    Try3_var1_post    Num       8 
                                  3    Try3_var2_post    Num       8 
                                  4    Try3_var3_post    Num       8 
 
 
 
 
Tool 4: Replace Suffix on Selected Variables 
    
We can use same approach of Tool 3, Using %SYSFUNC to retrieve the dataset and 
variable information and replace the existing suffix with new name. 
 
 
/* Replacing Suffix on Selected Variables */ 
 
%macro replacesuffix(lib,dsn,start,end,oldsuffix,newsuffix); 
proc contents data=&lib..&dsn.; 
title 'before renaming'; 
run; 
 
data temp; 

set &lib..&dsn.; 
run; 
 
%LET ds=%SYSFUNC(OPEN(temp,i)); 
%let ol=%length(&oldsuffix.); 
%do i=&start %to &end; 
 %let dsvn&i=%SYSFUNC(VARNAME(&ds,&i)); 
 %let l=%length(&&dsvn&i); 
 %let vn&i=%SUBSTR(&&dsvn&i,1,%EVAL(&l-&ol))&newsuffix.; 
%end; 
data &lib..&dsn.; 
set temp; 
  
%do i=&start %to &end; 
&&vn&i=&&dsvn&i; 
drop &&dsvn&i; 
%end; 
%let rc=%SYSFUNC(CLOSE(&ds)); 
proc contents data=&lib..&dsn.; 
title ' Replacing Suffix on Selected variables '; 
run; 
%mend replacesuffix; 
 
data e; 
set a; 
run; 
 
%replacesuffix(WORK,E,2,4,post,Try4); 
 
 
Partial OUTPUT of Tool 4: 
 
                              Replacing Suffix on Selected variables  

 
                             Alphabetic List of Variables and Attributes 
 
                                  #    Variable         Type    Len 
 
                                  1    id               Char      4 
                                  2    pre_var1_Try4    Num       8 
                                  3    pre_var2_Try4    Num       8 
                                  4    pre_var3_Try4    Num       8 
 
 
 
 

CAVEATS 
 
There are some caveats of using PROC SQL’s DICTIONARY.TABLES and 
COLUMNS. The library and dataset name must in Uppercase. Using the upcase function 
is highly recommended here. Besides that, only the prefix can be concatenated, which we 
need to find out a new way to concatenate the suffix. 

CONCLUSION 
 
As demonstrated above, the Renaming Tool Series perform different renaming jobs with 
hundreds of variables names in batches. Almost all the situations except adding the suffix 
can be handled. We currently are using a two step approach to adding the suffix; the code 
is in the Appendix. We still need to find a simple way to add the suffix and expect to add 
it to our toolbox soon.  

APPENDIX 
 
/* Adding Prefix/Suffix – Two Step Approach*/ 
 
/* 
*First, create two columns – varS, varE in Excel files. VarS is the 
original variable name, VarE contains the variables you want to change 
to. And read it into a SAS data set. 
*Then use the following code to perform the renaming, such as adding 
prefix.  
*/ 
 
%macro rename(lib,dsn, mthDs, varS,varE); 
proc contents data=&lib..&dsn; 
title 'before renaming'; 
run; 
 
proc sql noprint; 
 select count(*) into :num_vars 
 from &mthDs.; 
 
 select  &varS.,&varE. into :var1-:var%trim(%left(&num_vars)),:nvar1-
:nvar%trim(%left(&num_vars))  
 from &mthDs.; 
quit; 
run; 
 
proc datasets lib = &lib.; 
modify &dsn; 
rename 
%do i = 1 %to &num_vars.; 
&&var&i = &&nvar&i.  
%end; 
; 
quit; 
run; 
 
proc contents data=&lib..&dsn; 
title 'after renaming'; 
run; 
%mend rename; 

